import com.modrinth.minotaur.dependencies.ModDependency
plugins {
	id 'fabric-loom' version '1.11-SNAPSHOT'
	id 'maven-publish'
	id "com.modrinth.minotaur" version "2.+"
}

version = "${project.mod_version}+${project.minecraft_version}"
group = project.maven_group

base {
	archivesName = project.archives_base_name
}

repositories {
	mavenLocal()

	maven {
		name = 'ParchmentMC'
		url = 'https://maven.parchmentmc.org'
	}

	maven { url 'https://maven.nucleoid.xyz' }

	maven {
		name = "TerraformersMC"
		url = "https://maven.terraformersmc.com/"
	}
	maven {
		name = "Ladysnake Libs"
		url = 'https://maven.ladysnake.org/releases'
	}
}

loom {
	accessWidenerPath = file("src/main/resources/solstice.accesswidener")
}

dependencies {
	// To change the versions see the gradle.properties file
	minecraft "com.mojang:minecraft:${project.minecraft_version}"

	mappings loom.layered() {
		officialMojangMappings()
		parchment("org.parchmentmc.data:parchment-${project.minecraft_version}:${project.parchment_mappings}@zip")
	}

	modImplementation "net.fabricmc:fabric-loader:${project.loader_version}"

	modImplementation "net.fabricmc.fabric-api:fabric-api:${project.fabric_version}"

	include implementation("org.spongepowered:configurate-core:${project.configurate_version}")
	include implementation("org.spongepowered:configurate-hocon:${project.configurate_version}")
	include implementation("org.spongepowered:configurate-gson:${project.configurate_version}")
	include "com.typesafe:config:1.4.3"
	include "io.leangen.geantyref:geantyref:1.3.16"

	include modImplementation("me.lucko:fabric-permissions-api:${project.permissions_api_version}")

	include modImplementation("eu.pb4:placeholder-api:${project.placeholderapi_version}")
	include modImplementation("eu.pb4:sgui:${project.sgui_version}")
	modImplementation include("eu.pb4:common-economy-api:${project.commoneconomy_version}")

	modCompileOnly "dev.emi:trinkets:${project.trinkets_version}"
	modRuntimeOnly "dev.emi:trinkets:${project.trinkets_version}"

	modCompileOnly "net.luckperms:api:5.4"
	modRuntimeOnly "net.luckperms:api:5.4"
}

processResources {
	inputs.property "version", project.version

	filesMatching("fabric.mod.json") {
		expand "version": project.version
	}
}

tasks.withType(JavaCompile).configureEach {
	it.options.release = 21
}

java {
	// Loom will automatically attach sourcesJar to a RemapSourcesJar task and to the "build" task
	// if it is present.
	// If you remove this line, sources will not be generated.
	withSourcesJar()

	sourceCompatibility = JavaVersion.VERSION_21
	targetCompatibility = JavaVersion.VERSION_21
}

jar {
	from("LICENSE") {
		rename { "${it}_${project.base.archivesName.get()}"}
	}
}

modrinth {
	token = System.getenv("MODRINTH_TOKEN")
	projectId = 'uIvrDZas'
	uploadFile = remapJar
	gameVersions = [project.minecraft_version]
	loaders = ['fabric']
	dependencies = [
			new ModDependency('P7dR8mSH', 'required') // Fabric API
	]
}

// configure the maven publication
publishing {
	publications {
		create("mavenJava", MavenPublication) {
			artifactId = project.archives_base_name
			from components.java
		}
	}

	// See https://docs.gradle.org/current/userguide/publishing_maven.html for information on how to set up publishing.
	repositories {
		// Add repositories to publish to here.
		// Notice: This block does NOT have the same function as the block in the top level.
		// The repositories here will be used for publishing your artifact, not for
		// retrieving dependencies.


		maven {
			name = "AlexDevsRepo"
			url = "https://maven.alexdevs.me/releases"
			credentials {
				username = System.getenv("MAVEN_USERNAME")
				password = System.getenv("MAVEN_PASSWORD")
			}
		}
	}
}